
Problem Tanımlama 

 Problem kullanılacak denetleyiciye veya işlemciye göre teknik 

bir dille yeniden tanımlanır. 

 Derste verilen sorular genellikle tanımlanmış problemler 

olduğu için yeniden tanımlanmaz. 



Bağlantı Şekli 

23 Şubat 2015 Mikrodenetleyiciler 5 

 Eğer tanımlanan problem yeni bir bağlantı gerekiyorsa açık 

veya blok şema çizilir. 

 Özellikle bağlantı yapılacak port isimleri mutlaka yazılır. 

 Bağlanacak elemanların hatlarının isimleri yazılmalıdır. 



Algoritma 

 Çözüm yönteminin konuşma dili ile maddeler halinde 

tanımlanmasıdır. 

 Algoritma problemin çözümünü temel oluşturur. 

 Bağlantı şeklini desteklemelidir. 

 Yalın bir dil kullanılarak yazılmalıdır. 

 Bir problemin birden fazla çözüm algoritması olabilir. 



Akış Diyagramı 

 Algoritmanın şekilsel olarak gösterimidir. 

 Her yapılan işlem bir simge ile gösterilir. 

 Programın işleyiş sırasını anlamayı kolaylaştırır. 

İŞLEM 

ALT 
PROGRAM 

KARAR 

SONLANDIRMA 

VERİ 

MANUAL  
GİRİŞ 

MANUAL 

 İŞLEM 

DÖKÜMAN 

SAYFA BAŞI 

SAYFA SONU 



Kaynak Program 

 ;Programın başlığı, yazan kişi, tarih  

;program hakkında açıklama 

Org 0 ; program buradan başlar 

isim: Mov  A,#25 ;A=25 

……. ……. ;……………………….. 

TKR: ……. ……. ;……………………….. 

Sjmp $ ;son 

end ;program burada biter. 



Simulator 

 Yazılan program assebmly edilir. 

 Simulatöre yüklenerek programın çalışması adım modunda 

denetlenir. 

 Hata var ise algoritmaya geri dönülür. 

 Hatalar tamamen yok edilinceye kadar devam edilir 



Mikroişlemciye Yükleme ve Çalıştırma 

 Hataları giderilen program programlayıcı kullanılarak 
mikroişlemciye yüklenir. 

 Gerekli bağlantılar kart üzerinde ve çevresinde yapılır. 

 Devreye güç verilerek çalıştırılır ve sonuçlar gözlemlenerek 
doğruluğu belirlenir. 

 Hata var ise algoritmaya geri dönülür. 

 Hata yok ise tasarlanan cihaz üretime hazırdır. 



Örnek 

23 Şubat 2015 Mikrodenetleyiciler 11 

 Bir butona basıldığında LED’i yakan ikinci basmada ise 

söndüren programı yazın, sistemi kurup çalıştırın. 

1. Problemi Yeniden Tanımlama: 

 8051’in port hattına bir buton ve bir diğer hattına LED bağlıdır. 

Butona her basıldığında  LED’in durumunu tersleyen programı 

yazın. 

  



2. Bağlantı Şekli 

8051 

+5V 

10K 

1K 

S1 

P1.0 

P1.1 



Program Geliştirme 

3. Algoritma 4. Akış Diyagramı 

1. Buton için port hattını giriş yap. 

2. LED için port hattını çıkış yap 

3. LED’i başlangıç için söndür. 

4. Butona basılana kadar bekle. 

5. Buton bırakılana kadar bekle. 

6. LED’i tersle. 

7. Adım 4’e dön. 



5. Kaynak Program 

Org 0 

BUT_LED: 

 Mov P1,#0FFh    ;P1.0 giriş, P1.1 çıkış, LED sönük 

BUT: 

 Jb P1.0,$   ;Butona basılana kadar bekle 

 Jnb P1.0,$  ;Buton bırakılana kadar bekle 

 Cpl P1.1   ;LED’i tersle 

 Sjmp BUT ;Sürekli yap 

end 

 



Örnek 1 16 Bit Değişkene 8 Bit Değişken Toplama. 

Bu alt program A’yı DPTR’ye toplar, taşma var ise elde bayrağı 
kurulur. İşletilmeden önce DPTR’ye ve A’ya bir değer girilmelidir.  

 İşlem sonunda;  DPTR=DPTR+A 

 Etkilenen yazaçlar; PSW.CY, DPTR 

 
topla16b: 

 Push Acc ; A’yı yedekle. 
 Add A,DPL ; A’ya DPL topla. 
 Mov DPL,A ; Sonucu DPL’ye yaz. 
 Mov A,DPH ; DPH’yi al. 
 Addc A,#00h  ; Elde var ise YDB’ı artır. 
 Mov DPH,A ; DPH yaz. 
 Pop Acc  ; A’yı yedekten al. 
 Ret  ; Ana programa geri dön. 



ÖRNEK 2 16 Bit Değişkenden 8 Bit Değişken Çıkarma. 

;Bu alt program DPTR yazacından Acc’yi çıkarır. Sonuç negatif ise elde  
;bayrağını kurar. İşletilmeden önce DTPR’ye ve Acc’ye bir değer girilmelidir.  
;İşlem sonunda DPTR=DPTR-Acc 
;Etkilenen yazaçlar; PSW.CY, DPTR 

cikar16b: 
 Push Acc  ; A’yı yedekle. 
 Clr c   ; Eldeyi temizle 
 Xch a,DPL  ; A ile DPL’yi yer değiştir. 
 Subb a,DPL  ; Çıkar. 
 Mov DPL,a  ; DPL’yi yaz 
 Mov a,DPH  ; DPH’yi al. 
 Subb a,#00h  ; Borç var ise YD kısmı bir  azalt. 
 Mov DPH,a  ; DPH’yi yaz. 
 Pop Acc  ; A’yı yedekten al 
 Ret    ; Ana programa geri dön 



ÖRNEK 3  16 Bitlik Çıkarma 

;Bu program 16 bit sayıdan 16 bit sayıyı çıkarır.  

;R7R6-R5R4=R3R2 

Cıkar16bit: 

 MOV A,R6   ;1. değişkenin düşük kısmını A’ya al 

  CLR C      ;Eldeyi temizle 

   SUBB A,R4  ;2. değişkenin düşük kısmını A’dan çıkar  

  MOV R2,A   ;Sonucun düşük kısmını yaz 

 MOV A,R7   ;1. değişkenin yüksek kısmını A’ya al  

 SUBB A,R5  ;2. değişkenin yüksek kısmını A’dan çıkar 

 MOV R3,A   ;Sonucun yüksek kısmını yaz  

 RET  ;Sonucu R2 ve R3’te ana programa götür. 



ÖRNEK 4 DPTR Yazacını Bir Bit Sağa Öteleme. 

/* Bu alt program DPTR yazacının içeriğini aritmetik bir bit sağa öteler. 
İşletilmeden önce DPTR içerisine bir değer girilmeli. */ 

Dptrsag: 

  Push Acc  ; A’yı yedekle. 

  Clr C   ; Eldeyi temizle. 

  Mov A,DPH  ; DPTR’nin yüksek kısmını al. 

  Rrc A   ; Sağa ötele. 

  Mov DPH,A  ; DPH’ye yaz. 

  Mov A,DPL  ; DPTR’in düşük kısmını al. 

  Rrc A   ; Sağa ötele. 

  Mov DPL,A  ; DPL’yi yaz. 

  Pop Acc  ; Acc’yi tedekten al. 

  Ret    ; Ana programa dön. 



RLC A, RRC A   

Örnek 5: DPTR Yazacını Bir Bit Sola Öteleme. 

;Bu alt program DPTR yazacının içeriğini bir bit sola öteler.  

;DPTR * 2 = DPTR 

Dptrsol: 

  Push Acc  ; Acc’yi yedekle. 

  Clr C   ; Eldeyi temizle. 

  Mov A,DPL  ; DPTR’nin düşük kısmını al. 

  Rlc A   ; Sola ötele. 

  Mov DPL,A  ; DPL’ye yaz. 

  Mov A,DPH  ; DPTR’nin yüksek kısmını al. 

  Rlc A   ; Sola ötele. 

  Mov DPH,A  ; DPH’ye yaz.   

  Pop Acc  ;Yedekten al 

  Ret    ;  Ana programa dön. 



Örnek 6 DPTR’nin 10 Katını Alma. 

 ;Bu alt program DPTR’nin 10 katını alır. DPRTR’ye işlem öncesi bir 0-6553 arası bir değer 
yazılmalıdır. İşlem sonunda DPTR=DPTR * 10, etkilenen yazaçlar PSW,DPTR 
 
DPTRX10: 
 Push B  ; B’yi yedekle. 
 Push  Acc  ; A’yı yedekle 
 Mov A,DPL  ; Düşük kısmı al 
 Mov B,#10  ; Çarpanı B’ye yaz 
 Mul AB  ; BA = DPL * 10 
 Mov DPL,A  ;Sonucun düşük kısmını DPL yaz 
 Push  B  ;Sonucun yüksek kısmını yedekle 
 Mov A,DPH  ;Yüksek kısmını al 
 Mov B,#10  ;Çarpanı B’ye yaz 
 Mul AB  ;BA = DPH * 10 
 Pop B  ;B yedekten al 
 Add A,B  ;Düşük kısımdan oluşan ile topla 
 Mov DPH,A  ;Sonucun yüksek kısmını DPH yaz 
 Pop Acc  ;A’yı yedekten al 
 Pop b  ;B’yi yedekten al 
 ret   ;Ana programa dön 



Örnek 7 DPTR’yi 4 Bit Sağa Ötele 

;Bu Alt Program Dptr’yi 4 Bit Sağa öteler. Etkilenen Yazaçlar Psw,Dptr, DPTR’yi 
16’ya böler. 

Dptr4sag:  

  Push Acc ; A’yı Yedekle 

  Push 0  ; R0’ı Yedekle 

  Mov R0,#4 ; Öteleme Sayacını Ayarla 

L1:  Mov A,DPH ; Yüksek Kısmı Al 

  Clr C 

  Rrc A  ; Sağa ötele 

  Mov DPH,A ; DPH’ye Yaz 

  Mov A,DPL ; Dptr’nin düşük Kısmını Al 

  Rrc A  ; Sağa ötele 

  Mov DPL,A 

  Djnz R0,L1 ; Sayaç=0 olana kadar ötele 

   Pop 0  ; Yedekten Al 

  Pop Acc  ; Yedekten Al 

  Ret  ; Ana Programa Geri Dön 



Örnek 8 Dptr 4 Bit Sola Öteleme 

;Bu Alt Program Dptr Yazacını 4 Bit Sola Öteler, Düşük Değerli Bitlere Sıfır Yerleştirir. 
;Etkilenen Yazaçlar; Psw.Cy, Dptr 
Dptr4sol: 
  Push Acc ; A’yı Yedekle 
  Push 0  ; R0’ı Yedekle 
  Mov R0,#4 ; Öteleme Sayacını Kur 
L2: Clr C ; Öteleme İçin Sıfırla 
  Mov A,Dpl ; Dptr’nin Düşük Kısmını Al 
  Rlc A ; Sola Ötele 
  Mov Dpl,A ; Dpl’yi Yaz 
  Mov A,Dph ; Dptr’nin Yüksek Kısmını Al 
  Rlc A ; Sola Ötele 
  Mov Dph,A ; Dph’yi Yaz 
  Djnz R0,L2  ; Sayaç Sıfır Olana Kadar Devam Et 
  Pop 0 ; Yedekten Al 
  Pop Acc ; Yedekten Al 
  Ret  ; Ana Programa Geri Dön 



Örnek 9  AND Maskeleme ve Tablo Okuma 

 P1’den okunan 0 ile F arası sayıların karesini başvuru tablosu kullanarak 
alan ve P2’ye yazan programı yazın 

Kare_al:      

  Mov DPTR, #TABLO ; Tablonun başlangıç adresi 

  Mov P1, #0FFH  ; P1’i giriş yap 

TKR: Mov A, P1   ; Sayıyı P1’den oku 

  Anl A,#0Fh  ;Yüksek değerli 4 biti sıfırla 

  Movc A, @A+DPTR  ; Sayının karesini tablodan al 

  Mov P2, A   ; Sonucu P2’ye yaz 

  Sjmp TKR   ; Sürekli yap 

ORG 300H    ;Tablo 0300H’dan başlar 

TABLO: DB  0, 1, 4, 9, 16, 25, 36, 49, 4,81,100,121,144,169, 
196,225 



8051’de Yığın Kullanımı 

23 Şubat 2015 Mikrodenetleyiciler 25 

 Yığını SP (stack pointer) 
yazacı ile erişebiliriz. 

 Yığın işaretleyici SP 
8051’de 8 bit 
genişliğindedir, bunun 
anlamı bu yazaç içeriği  
00’dan FFH kadar 
değerleri içerebilir.  

 8051 çalıştırıldığında, SP 
yazacının içeriği  07’dir. 08h-0Fh 

Bank 3 

Bit Adreslenebilir 

RAM 

Genel  RAM 

Bank 2 

Bank 1 

Bank 0 00h-07h 

10h-17h 

18h-1Fh 

20h-2Fh 

30h-7Fh 



Yığın???? 

Altprogramların Kullanımı 

ANA_PROG: 

Org 0 

 Mov SP,#7Fh  ;yığını üst 128 bayta ötele 

 Mov A,#25H  ;A=25H 

 Acall topla16b  ;toplama altprogramını çağır 

 Acall Cikar16bit  ;Çıkarma altprogramını çağır 

 Mov DPTR,#2551  ;DPTR=2551 

 Acall DPTRX10  ;DPTR*10 

 Acall Dptr4sol  ;DPTR*2*2*2*2 

 Acall Cikar16bit  ;Çıkarma altprogramını çağır 

 Sjmp $ 

 

 

 

 

  



Ödev - 4 

 Bölüm 4’ün sorularını yanıtlayınız. 

 (Haftaya ders saatinde teslim edilecek) 


